

# Modèles multi-espèces de pêche

## Les questions

Michel DE LARA et Mohamed SBAI

October 10, 2017

### Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Le modèle SSVPA (Single Species Virtual Population Analysis)</b>	<b>1</b>
<b>3</b>	<b>Simulation pour une population de hareng</b>	<b>3</b>
<b>4</b>	<b>Le modèle MSVPA (Multi Species Virtual Population Analysis)</b>	<b>4</b>
<b>5</b>	<b>Simulation pour une population de merlu-anchois</b>	<b>7</b>

## 1 Introduction

L'approche écosystémique de la gestion des pêches est en plein développement et il est de plus en plus nécessaire de disposer de modèles de poissons fiables et qui rendent bien compte de l'évolution des populations marines.

On s'intéresse ici aux modèles multi-espèces de poissons exploités par la pêche. Actuellement, ces modèles sont principalement utilisés dans la mer du Nord et la mer Baltique.

## 2 Le modèle SSVPA (Single Species Virtual Population Analysis)

Développé par Gulland (1965), ce modèle structuré en âge ne rend pas compte des interactions entre différentes espèces.

On représente par  $N_{a,t}$  le nombre d'individus d'âge  $a$  au début de la période  $[t, t + 1[$

(généralement une année ou un trimestre) et  $C_{a,t}$  le nombre d'individus capturés (par la pêche) sur la période  $[t, t+1[$ . Les 2 équations fondamentales de la SSVPA sont :

$$C_{a,t} = N_{a,t} \frac{F_{a,t}}{Z_{a,t}} (1 - e^{-Z_{a,t}}) \quad (1)$$

$$N_{a+1,t+1} = N_{a,t} e^{-Z_{a,t}} \quad (2)$$

avec  $Z_{a,t}$  la mortalité entre  $t$  et  $t + 1$  des individus d'âge  $a$ . On a :

$$Z_{a,t} = F_{a,t} + M_a \quad (3)$$

où

- $M_a$  est la mortalité naturelle, donnée extérieure au problème;
- $F_{a,t}$  est la mortalité due à la pêche entre  $t$  et  $t + 1$ .

Pour évaluer le nombre  $N_{0,t+1}$  d'individus d'âge 0 au début de la période  $[t + 1, t + 2[$ , on utilise la relation de stock-recrutement de Ricker :

$$N_{0,t+1} = \alpha SSB_{t+1} e^{-\beta SSB_{t+1}}, \quad (4)$$

avec

- $\alpha$  et  $\beta$  : paramètres de la relation stock-recrutement de Ricker (constantes) ;
- $SSB_{t+1}$  : la biomasse féconde en fin de période  $[t, t+1[$  calculée en sommant la contribution de tous les individus à la reproduction (sauf ceux d'âge 0) :

$$SSB_{t+1} := \sum_{a=1}^A w_a N_{a,t+1} P_a = \sum_{a=1}^A w_a N_{a-1,t} e^{-Z_{a-1,t}} P_a, \quad (5)$$

où

- $w_a$  est le poids aux âges
- $P_a$  est la proportion de reproducteurs aux âges

On veut écrire ce modèle sous forme de modèle d'état. Le vecteur d'état qu'on prend représente le stock :

$$x(t) = \begin{pmatrix} N_{0,t} \\ \vdots \\ N_{A,t} \end{pmatrix}$$

et le vecteur commande l'effort de pêche :

$$u(t) = \begin{pmatrix} F_{0,t} \\ \vdots \\ F_{A,t} \end{pmatrix}$$

A partir de (2) et (4) on trouve que :

$$x(t+1) = \begin{pmatrix} g\left(\sum_{a=1}^A w_a P_a(N_{a-1,t} \exp(-(M_{a-1} + F_{a-1,t})))\right) \\ N_{0,t} \exp(-(M_0 + F_{0,t})) \\ N_{1,t} \exp(-(M_1 + F_{1,t})) \\ \vdots \\ N_{A-1,t} \exp(-(M_{A-1} + F_{A-1,t})) \end{pmatrix} \quad (6)$$

avec

$$g(z) = \alpha z e^{-\beta z}.$$

Il s'agit bien d'une équation dynamique de la forme

$$x(t+1) = f(x(t), u(t)).$$

### 3 Simulation pour une population de hareng

Dans cette partie, on va simuler l'évolution d'une population de hareng de la mer du Nord sur un horizon fini.

Le hareng est à dix classes d'âge. Le vecteur d'état qu'on prend est :

$$x(t) = \begin{pmatrix} N_{0,t} \\ N_{1,t} \\ \vdots \\ N_{9,t} \end{pmatrix}$$

D'après ce qui précède, on a l'équation dynamique suivante :

$$x(t+1) = \begin{pmatrix} g\left(\sum_{a=1}^9 w_a P_a(N_{a-1,t} \exp(-(M_{a-1} + F_{a-1,t})))\right) \\ N_{0,t} \exp(-(M_0 + F_{0,t})) \\ N_{1,t} \exp(-(M_1 + F_{1,t})) \\ \vdots \\ N_{8,t} \exp(-(M_8 + F_{8,t})) \end{pmatrix}$$

avec  $g(z) = \alpha z e^{-\beta z}$ .

Ouvrir un fichier `ssvpa.sce` et y recopier les données numériques suivantes:

```

//etat initial
X0=[59136429;15134341;16157668;12683928;6271295;3193827;1317572;453613;298854;101254];
//poids aux âges
W=[0.007;0.023;0.035;0.051;0.065;0.059;0.075;0.079;0.087;0.094];
//Proportion mature
P=[0.00;0.00;0.70;0.90;1.00;1.00;1.00;1.00;1.00;1.00];
//mortalité par pêche constante
F=[0.00;0.05;0.14;0.21;0.45;0.30;0.38;0.40;0.39;0.39];
//coeff de Ricker
alpha=140.9725694;bbeta=0.00000081;
//mortalité naturelle
M=0.2;
//horizon fini
T=15;

```

On écrit la fonction de la dynamique. Pour cela, on écrit d'abord la fonction qui donne la biomasse féconde.

```

//biomasse féconde
function y=Bf(x)
    y=sum(P .*W .*x)
endfunction
// on a P(1)=0, de sorte que la somme ne porte que sur les individus matures

//fonction dynamique
function y=dyn(k,x)
    e=exp(-M-F);
    y(2:10)=x(1:$-1) .*e(1:$-1);
    z=[0;y(2:$)];
    y(1)=alpha*Bf(z)*exp(-bbeta*Bf(z));
endfunction

```

Dans notre modèle, la biomasse  $B$  d'une espèce donnée s'écrit :

$$B = \sum_{a=0}^A w_a N_{a,t}. \quad (7)$$

**Question 1** Représenter l'évolution de la biomasse. On utilisera la commande `plot2d2`.

```

y=ode("discrete",X0,1,1:T,dyn);
B=W'*y;// biomasse

```




Figure 1: Evolution de la biomasse du hareng

```

xbasc(0);
xset("window",0)
plot2d2(1:T,B)
xtitle('biomassehareng','temps(annee)','biomasse(kg)')

```

**Question 2** Répéter la même opération pour différentes valeurs de l'horizon ( $T=20$  et  $T=50$ ) et de l'effort de pêche. Commenter.

## 4 Le modèle MSVPA (Multi Species Virtual Population Analysis)

Cette fois, on intègre les interactions entre différentes espèces dans la modélisation en distinguant dans la mortalité naturelle un terme dû à la prédation. On garde les mêmes notations en ajoutant l'indice  $i$  ( $i = 1, \dots, n$ ) pour distinguer entre les différentes espèces:

$$\begin{aligned} N_{i,a+1,t+1} &= N_{i,a,t} e^{-Z_{i,a,t}} \\ N_{i,0,t} &= g_i(SSB_{i,t}) \end{aligned} \quad (8)$$

où  $g_i$  est la fonction de recrutement pour l'espèce  $i$ , par exemple (mais pas forcément) la relation de stock-recrutement de Ricker

$$g_i(z) := \alpha_i z e^{-\beta_i z}$$

et avec

$$Z_{i,a,t} = F_{i,a,t} + M_{1,i,a} + M_{2,i,a,t}$$

où

- $F_{i,a,t}$  est la mortalité due à la pêche pour l'espèce  $i$  entre  $t$  et  $t + 1$ ;
- $M_{2,i,a,t}$  est la mortalité par prédation entre  $t$  et  $t + 1$ ;
- $M_{1,i,a}$  est la mortalité "résiduelle" pour l'espèce  $i$ , englobant toutes les autres causes de mortalité et prise comme donnée extérieure au problème comme en SSVPA.

En prenant un vecteur d'état de la forme

$$x^i(t) = \begin{pmatrix} N_{i,0,t} \\ \vdots \\ N_{i,A,t} \end{pmatrix}$$

comme en **SSVPS** on ne peut pas avoir un modèle d'état.

On prend alors pour vecteur d'état le vecteur de vecteurs

$$x(t) = \begin{pmatrix} x^1(t) \\ \vdots \\ x^n(t) \end{pmatrix}$$

Pour l'espèce  $i$ , on a d'après (8)

$$x^i(t+1) = \begin{pmatrix} g_i \left( \sum_{a=1}^A w_{i,a} P_{i,a} N_{i,a-1,t} \exp(-Z_{i,a-1,t}) \right) \\ N_{i,0,t} \exp(-Z_{i,0,t}) \\ N_{i,1,t} \exp(-Z_{i,1,t}) \\ \vdots \\ N_{i,A-1,t} \exp(-Z_{i,A-1,t}) \end{pmatrix} \quad (9)$$

Par rapport au modèle précédent, le nouveau terme  $M_{2,i,a,t}$  couple les différentes équations d'état étant donné qu'il s'exprime en fonction de la biomasse de toutes les autres espèces. En effet, pour un prédateur, on a  $M_2 = 0$  et pour une proie  $p$  d'âge  $b$  on a :

$$M_{2,p,b,t} = \sum_{i \in \mathcal{P}} \sum_a \frac{S(i, a : p, b) R(i, a, t) N_{i,a,t}}{BS(i, a, t)}$$

avec

- $\mathcal{P}$  l'ensemble des prédateurs;
- $S(i, a : p, b)$  le coefficient qui dénote la convenance d'une proie  $p$  d'âge  $b$  en tant que nourriture pour un prédateur  $i$  d'âge  $a$ ;
- $R(i, a, t)$  la quantité d'aliments consommés au temps  $t$  par le prédateur  $i$  d'âge  $a$ .

- $BS(i, a, t)$  la biomasse de toutes les proies d'un prédateur  $i$  d'âge  $a$  au temps  $t$ ; si on note par  $\mathcal{A}$  l'ensemble des proies alors :

$$BS(i, a, t) = S(i, a : ex)B_{ex} + \sum_{p \in \mathcal{A}} \sum_b S(i, a : p, b)w_{p,b}N_{p,b,t}$$

où

- $B_{ex}$  est la biomasse externe, c'est à dire la biomasse totale de l'écosystème privée de la biomasse des espèces incluses dans le modèle.
- $S(i, a : ex)$  est le coefficient qui dénote la convenance de la biomasse externe en tant que nourriture pour le prédateur  $i$  d'âge  $a$

On pose  $u^i(t) = \begin{pmatrix} F_{i,0,t} \\ \vdots \\ F_{i,A,t} \end{pmatrix}$  et on définit le vecteur commande par  $u(t) = \begin{pmatrix} u^1(t) \\ \vdots \\ u^n(t) \end{pmatrix}$

Pour simplifier, on suppose que  $R(i, a, t)$  est indépendante du temps.

On développe l'expression de  $M2$ ; on trouve que pour une espèce  $i$ , la mortalité est de la forme :

$$Z_{i,a,t} = \psi_{i,a}(x(t), u^i(t))$$

En posant

$$\phi_{i,a}(x(t), u^i(t)) = \exp(-\psi_{i,a}(x(t), u^i(t)))$$

l'équation (9) devient

$$x^i(t+1) = \begin{pmatrix} g_i \left( \sum_{a=1}^A w_{i,a} P_{i,a} x_{a-1}^i(t) \phi_{i,a-1}(x(t), u^i(t)) \right) \\ x_0^i(t) \phi_{i,0}(x(t), u^i(t)) \\ x_1^i(t) \phi_{i,1}(x(t), u^i(t)) \\ \vdots \\ x_{A-1}^i(t) \phi_{i,0}(x(t), u^i(t)) \end{pmatrix} \quad (10)$$

Soit  $h_{i,a}$  la fonction

$$h_{i,a}(x, u) := x_a^i \exp(-\phi_{i,a}(x, u^i)). \quad (11)$$

On peut écrire que

$$x^i(t+1) = \begin{pmatrix} g_i \left( \sum_{a=1}^A w_{i,a} P_{i,a} h_{i,a-1}(x(t), u(t)) \right) \\ h_{i,0}(x(t), u(t)) \\ h_{i,1}(x(t), u(t)) \\ \vdots \\ h_{i,A-1}(x(t), u(t)) \end{pmatrix} = f_i(x(t), u).$$

Finalement on obtient une équation dynamique de la forme voulue

$$x(t+1) = \begin{pmatrix} x^1(t+1) \\ \vdots \\ x^n(t+1) \end{pmatrix} = \begin{pmatrix} f_1(x(t), u(t)) \\ \vdots \\ f_n(x(t), u(t)) \end{pmatrix} = f(x(t), u(t)).$$

## 5 Simulation pour une population de merlu-anchois

En s'inspirant de la population marine du Golfe de Gascogne, on considère l'écosystème suivant composé d'un prédateur et d'une proie :

- merlu (prédateur) à 3 classes d'âge; on adopte la notation  $N_{a,t}^m$ ,  $a = 0, 1, 2$  pour désigner le nombre d'individus d'âge  $a$  au temps  $t$ ;
- anchois (proie) à 2 classes d'âge ( $N_{a,t}^{an}$ ,  $a = 0, 1$ ).

Le vecteur d'état est :

$$x(t) = \begin{pmatrix} N_{0,t}^m \\ N_{1,t}^m \\ N_{2,t}^m \\ N_{0,t}^{an} \\ N_{1,t}^{an} \end{pmatrix}$$

On suppose que la mortalité par pêche dépend de deux variables de contrôle  $u(t)$  et  $v(t)$  :

- le multiplicateur de mortalité par pêche des merlus  $F_{a,t}^m = u(t)F_a^m$ ;
- le multiplicateur de mortalité par pêche des anchois  $F_{a,t}^{an} = v(t)F_a^{an}$ .

D'après ce qui précède qu'on aboutit à l'équation dynamique

$$x(t+1) = \begin{pmatrix} g^m(p_1^m w_1^m N_{0,t}^m \exp(-M^m - u(t)F_0^m) + p_2 w_2 N_{1,t}^m \exp(-M^m - u(t)F_1^m)) \\ N_{0,t}^m \exp(-M^m - u(t)F_0^m) \\ N_{1,t}^m \exp(-M^m - u(t)F_1^m) \\ g^{an}(p_1^{an} w_1^{an} N_{1,t}^{an} \exp(-M^{an} - v(t)F_0^{an} - M_2)) \\ N_{0,t}^{an} \exp(-M^{an} - v(t)F_0^{an} - M_2) \end{pmatrix}$$

avec

$$M_2 = \sum_{a=0}^2 \frac{r_a s_{a,0} N_{a,t}^m}{s_{a,ex} Bex + s_{a,0} w_0^m N_{0,t}^m + s_{a,1} w_1^m N_{1,t}^m}.$$

Elle est de la forme  $x(t+1) = f(x(t), u(t), v(t))$ .

On suppose  $u(t)$  et  $v(t)$  constants. Pour le merlu (prédateur), on utilise la relation de stock-recrutement de Ricker

$$g^m(z) := \alpha^m z e^{-\beta^m z} \quad (12)$$

mais, pour l'anchois, on n'utilise pas la relation de stock-recrutement de Ricker. En effet, l'anchois est une espèce qui est très sensible aux fluctuations environnementales (climat) à court terme ; on considère que le recrutement varie aléatoirement autour d'une valeur moyenne :

$$g^m(z) = \bar{g}^m + \text{aléa.} \quad (13)$$

Charger dans un fichier `merlu_anchois.sce` les données numériques suivantes

```
//etat initial
X0=[82764000;26548000;41440000;18220000;4264000];
//poids aux ages
W=[0.004;0.037;0.106;0.012;0.016];
//proportion mature
Pm=[0;0;1;0;1];
//mortalite residuelle
M1=0.2;
//quantite d'aliments manges par les merlus aux ages
R=[1;1.2;1.4];
//coeff de la relation de Ricker
alpham=8.745;betam=0.00000000143;
//valeur moyenne de recrutement pour les anchois
recrumoya=1500000000;
//tableau des coef. de suitability
suit=[0,0,1;0.25,0.15,0.6;0.185,0.385,0.43];
F1=0.2;
F2=0.3;
F4=0.5;
//biomasse externe
Bex=10000000;
//multiplicateurs de mortalité
u=[1.22;2.35;0.856];
```

Recopier dans le même fichier les fonctions de mortalité par pêche  $M_2$  et de biomasse féconde pour le merlu.

```

//mortalité par pêche
function y=M2(x)
    y=(suit(1,1)*R(1)*x(1))/(suit(1,3)*Bex+suit(1,1)*W(4)*x(4)+suit(1,2)*W(5)*x(5))+ ...
        (suit(2,1)*R(2)*x(2))/(suit(2,3)*Bex+suit(2,1)*W(4)*x(4)+suit(2,2)*W(5)*x(5))+ ...
        (suit(3,1)*R(3)*x(3))/(suit(3,3)*Bex+suit(3,1)*W(4)*x(4)+suit(3,2)*W(5)*x(5))
endfunction

//biomasse féconde pour le merlu
function y=Bfm(x)
    y=Pm(2)*W(2)*x(2)+Pm(3)*W(3)*x(3)
endfunction

```

On écrit à la suite la fonction qui donne la dynamique  $f$  du modèle merlu\_anchois.

```

//dynamique
function y=dyn(k,x)
    y(2)=x(1)*exp(-M1-u(1)*F1)
    y(3)=x(2)*exp(-M1-u(2)*F2)
    y(1)=alphan*Bfm(y)*exp(-betam*Bfm(y))
    y(4)=recrumoya*(1+0.5*(rand()-0.5))
    // y(4)=300*x(5)*W(5)
    y(5)=x(4)*exp(-M1-u(3)*F4-M2(x))
endfunction

```

**Question 3** Représenter l'évolution de la biomasse de chacune des deux espèces avec la commande `plot2d2`.

```

y=ode("discrete",X0,1,1:T,dyn);
Bm=W(1:3)'*y(1:3,:);
xbasec(0);
xset("window",0)
plot2d2(1:T,Bm)
xtitle('biomassemerlu','temps(annee)','biomasse(kg)')

Ba=W(4:5)'*y(4:5,:);
xbasec(1);
xset("window",1)
plot2d2(1:T,Ba)
xtitle('biomasseanchois','temps(annee)','biomasse(kg)')

```

**Question 4** Répéter la même opération pour différentes valeurs de  $u$  et de l'horizon ( $T=20$  et  $T=30$ ). Commenter.




Figure 2: Evolution de la biomasse des merlus




Figure 3: Evolution de la biomasse des anchois

## References

- [1] Kjartan G.Magnusson, *An overview of the multispecies VPA - theory and applications*, Reviews in Fish Biology and Fisheries, 5 195-212 (1995).
- [2] H.Gislason *Single and Multispecies reference points for Baltic fish stocks*, ICES Journal of Marine Science, 56: 571-583. 1999
- [3] H.Gislason, Th.Helgason, *Species interaction in assesment of fish stocks with special application to the North Sea*, Dana, vol.5, pp.1-44, 1985.
- [4] J. G.Pope, *The ICES Multispecies Assesment Working Group: evolution, insights and future problems*, ICES mar.Sci.Symp.,193:22-33. 1991.
- [5] *Report of the Workshop on MSVPA in the North SEA*, Charlottenlund, Denmark 8-12 avril 2002.
- [6] Sparre.P, *Introduction to multispecies virtual population analysis*, ICES mar.Sci.Symp.,193:12-21.1991.
- [7] Mark Kot, *Elements of Mathematical Ecology*, Cambridge University Press.